Newsgroup: it.comp.software.newsreader
Titolo: [Xnews] launcher con Mime-Proxy

Tempo fa mi sembra che *Alan Ford* avesse scritto un programma/script 
che permetteva di lanciare e terminare Xnews e Mime-Proxy in modo molto 
semplice.
Mi ricordo che la discussione era nata dalla necessità di avere una 
versione di Xnews portabile.
Dato che non trovo il link, qualcuno potrebbe ricordarmelo?
-- 
Ciao, Lurkos
it-alt.media.internet.google
I servizi di Google: Gmail, Groups, News ed altro

 semplice.
non so cosa avesse scritto Alan, ma basta un semplice file .bat eh
-- 
Ciao, Dmitry
X-Privat F.A.Q. : 
http://www.x-privat.org/faq.php

 
Grazie, era proprio quello.
Vado a scaricare XnewsGo.


 
XnewsGo! termina anche Mime-Proxy alla chiusura di Xnews?
Dalle prove che ho fatto sembra di no.


 XnewsGo! termina anche Mime-Proxy alla chiusura di Xnews?
No, all'avvio di Xnews si termina da solo (ma a cosa ti serve, se lo
avvii senza console puoi killarlo dai task facendo Ctrl+Alt+Canc).
XnewsGo! è nato per gioco (prevalentemente per togliere le tracce dal
PC ospite e per prendere confidenza con gli script di AutoIt3).
Mi pareva di aver messo anche il sorgente, nel caso te lo posto dopo la
firma con la modifica per chiuderlo (sempre il primo tasto, dopo la
domanda), oppure potresti avviarlo senza l'opzione -free_console, così
clicchi semplicemente sulla X, fai tu.
Ciao.
-- 
 /   _ _ | Xnews e la documentazione in italiano
( (/( (/ | 
http://web.lacasadialice.it/alanford/
----- 8< ----- 8< -----
; Xnews Go! è un launcher per Xnews scritto con AutoIt3
#include <GUIConstants.au3>
GUICreate("Xnews Go! 1.2",200,60)
$button1 = GUICtrlCreateButton ("1", 0,0,40,40,$BS_ICON)
GUICtrlSetImage (-1, ".\mproxy\mproxy.ico",0)
GUICtrlSetTip(-1, "Avvia(termina) Mime-proxy e Xnews")
$button2 = GUICtrlCreateButton ("2", 40,00,40,40,$BS_ICON)
GUICtrlSetImage (-1, "xnews.exe",0)
GUICtrlSetTip(-1, "Avvia solamente Xnews")
$button3 = GUICtrlCreateButton ("3", 160,00,40,40,$BS_ICON)
GUICtrlSetImage (-1, "shell32.dll",31)
GUICtrlSetTip(-1, "Cancella le tracce di Xnews dal Registro di questo PC")
$button4 = GUICtrlCreateButton ("4", 80,0,40,40,$BS_ICON)
GUICtrlSetImage (-1, "shell32.dll",44)
GUICtrlSetTip(-1, "Salva le info del Registro su chiavetta")
$button5 = GUICtrlCreateButton ("5", 120,0,40,40,$BS_ICON)
GUICtrlSetImage (-1, "shell32.dll",12)
GUICtrlSetTip(-1, "Scrivi nel Registro le impostazioni salvate sulla chiavetta")
GUISetState()
$trackmenu = GuiCtrlCreateContextMenu ()
$aboutitem = GuiCtrlCreateMenuitem ("About",$trackmenu)
$aboutlabel = GUICtrlCreateLabel ("by Alan Ford 2005",2,45,100,20)
; Run the GUI until the dialog is closed or enconter an ExitLoop
While 1
	$msg = GUIGetMsg()
        If $msg = $button1 Then
          If ProcessExists("mproxy.exe") = 0 Then 
              If FileExists(".\mproxy\mproxy.exe") = 1 AND FileExists(".\mproxy\mproxy.ini") = 1 Then
                Run(".\mproxy\mproxy.exe .\mproxy\mproxy.ini -low_memory -free_console")
                Sleep(6000)
                Run("xnews.exe")
                ExitLoop
              Else
                MsgBox(48, "Errore", "File di Mime-proxy mancanti! ")
              EndIf        
          Else
            MsgBox(48, "Errore", "Mime-Proxy risulta già in esecuzione. ")
            $flag = MsgBox(33, "Domanda", "Si desidera terminarlo? ")
            If $flag = 1 Then ProcessClose ("mproxy.exe")
          EndIf
        EndIf
        If $msg = $button2 Then
            Run("xnews.exe")
            ExitLoop
        EndIf
        If $msg = $button3 Then
          $flag = MsgBox(33, "Conferma", "Sto per rimuovere la chiave di Registro relativa ad Xnews su questo PC. Procedere? ")
          If $flag = 1 Then
            RegDelete ("HKEY_CURRENT_USER\Software\Xnews")
            MsgBox(48, "Avviso", "Chiave di Registro cancellata. ")
          EndIf 
        EndIf
        If $msg = $button4 Then
          Run("regedit /e xn.reg HKEY_CURRENT_USER\Software\Xnews")
          MsgBox(48, "Avviso", "Chiave di Registro aggiornata nel file xn.reg ")
        EndIf
        If $msg = $button5 Then
          If FileExists("xn.reg") = 1 Then
            $flag = MsgBox(33, "Conferma", "Sto per inserire le informazioni contenute nella chiavetta relative ad Xnews, nel Registro di questo PC. È importante che siano state precedentemente salvate da un PC con il medesimo Sistema Operativo. Procedere? ")
            If $flag = 1 Then
              Run("regedit /s xn.reg")
              MsgBox(48, "Avviso", "Informazioni inserite nel Registro. ")
            EndIf
          Else
            MsgBox(48, "Errore", "File con chiave di registro assente! ")
          EndIf 
        EndIf
        If $msg = $aboutitem Then Msgbox(0,"About","Launcher da chiavetta per Xnews. Copiare la cartella di Xnews sulla chiavetta, rendere 'relativi' i percorsi nel file Xnews.ini; se si copia anche il Mime-proxy metterlo nella sottocartella 'Mproxy' di Xnews ed assicurarsi che il file di configurazione si chiami mproxy.ini. Non rispondo di eventuali danni sul vostro PC: usatelo a vostro rischio e pericolo! ")
        If $msg = $GUI_EVENT_CLOSE Then ExitLoop
Wend
GUIDelete()
----- 8< ----- 8< -----